Quadriken 3d Ellipsoid

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Juni 07   Update 5.07.07
Web: http://haftendorn.uni-lueneburg.de   www.mathematik-verstehen.de

##

Rückwärts aufgestellt, siehe Datei Konstruktion-rückwärts

```
p:=matrix([x,y,z]):pt:=linalg::transpose(p)
```

$\begin{pmatrix} x & y & z \end{pmatrix}$

```
quadrik:=matrix([[7*x^2 - 4*x*y - 2*x*z - 70*x + 7*y^2 -
2*y*z + 38*y + 4*z^2 + 28*z + 202]])
```

$$\left(7\cdot x^2 - x\cdot y\cdot 4 - x\cdot z\cdot 2 - x\cdot 70 + 7\cdot y^2 - y\cdot z\cdot 2 + 38\cdot y + 4\cdot z^2 + 28\cdot z + 202 \right)$$

```
quadrikp:=plot::Implicit3d(quadrik[1],x=-2..7,y=-5..0,z=
-7..0,FillColor=[0,1,0,0.8], Scaling=Constrained):
plot(quadrikp)
```



A und a passend aufstellen

```
A:=matrix([[7,-2,-1],[-2,7,-1],[-1,-1,4]]);
a:=matrix([-70,+38,28]): at:=linalg::transpose(a);
d:=202;
```

$$\begin{pmatrix} 7 & -2 & -1 \\ -2 & 7 & -1 \\ -1 & -1 & 4 \end{pmatrix}$$

$\begin{pmatrix} -70 & 38 & 28 \end{pmatrix}$

202

```
expand(pt*A*p+at*p+d);  //Probe, ob man A,a und d
richtig hat
%-quadrik
```

1

$$\left(7\cdot x^2 - x\cdot y\cdot 4 - x\cdot z\cdot 2 - x\cdot 70 + 7\cdot y^2 - y\cdot z\cdot 2 + 38\cdot y + 4\cdot z^2 + 28\cdot z + 202 \right)$$

$$\left(7 \cdot x^2 - x \cdot y \cdot 4 - x \cdot z \cdot 2 - x \cdot 70 + 7 \cdot y^2 - y \cdot z \cdot 2 + 38 \cdot y + 4 \cdot z^2 + 28 \cdot z + 202 \right)$$

$$(0)$$

hier muss 0 herauskommen

Hauptachsentransformation

```
E3:=matrix([[1,0,0],[0,1,0],[0,0,1]])
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
evli:=linalg::eigenvectors(A) //Probe, was MuPAD liefert
```

$$\left[\left[6, 1, \left[\begin{pmatrix} -1 \\ -1 \\ 1 \end{pmatrix} \right] \right], \left[3, 1, \left[\begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \\ 1 \end{pmatrix} \right] \right], \left[9, 1, \left[\begin{pmatrix} -1 \\ 1 \\ 0 \end{pmatrix} \right] \right] \right]$$

Eigenwerte und Eigenvektoren

```
ew1 :=evli[1][1];  ew2 :=evli[2][1]; ew3 :=evli[3][1];
ev1:=evli[1][3][1];
ev2:=evli[2][3][1];
ev3:=evli[3][3][1];
```

$$6$$

$$3$$

$$9$$

$$\begin{pmatrix} -1 \\ -1 \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2} \\ \frac{1}{2} \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} -1 \\ 1 \\ 0 \end{pmatrix}$$

```
ev1n:=linalg::normalize(ev1):
ev2n:=linalg::normalize(ev2):
ev3n:=linalg::normalize(ev3):
P:=ev1n.ev2n.ev3n: Pt:=linalg::transpose(P);
```


$$\left(\begin{matrix} -\frac{\sqrt{3}}{3} & -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \\ \frac{}{\sqrt{2} \cdot \sqrt{3}} & \frac{}{\sqrt{2} \cdot \sqrt{3}} & \frac{}{\sqrt{2} \cdot \sqrt{3}} \end{matrix} \right.$$

$$\begin{pmatrix} -\frac{\sqrt{3}}{3} & -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \\ \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{3} \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \end{pmatrix}$$

Vektorschreibweise für die Abbildung $\vec{p} = P\,\vec{p}\,'$ und die Quadrikgleichungen, die sich durch Einsetzen ergeben:

$$Q: \quad \vec{p}^{\,T} A\,\vec{p} + \vec{a}^{\,T}\vec{p} + d = 0$$

$$Q': \quad \vec{p}\,'^{T} P^{T} A\, P\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

$$Q': \quad \vec{p}\,'^{T} D_{EW}\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

mit $D_{EW} = \begin{pmatrix} \lambda_1 & 0 \\ 0 & \lambda_2 \end{pmatrix}$

bzw. in 3D

```
quastrich:=Simplify(pt*Pt*A*P*p+at*P*p+d)
```

$$\left(6\cdot x^2 + 20\cdot\sqrt{3}\cdot x + 3\cdot y^2 + 4\cdot\sqrt{6}\cdot y + 9\cdot z^2 + 54\cdot\sqrt{2}\cdot z + 202\right)$$

```
quastrichp:=plot::Implicit3d(quastrich[1],x=-5..0,y=-5..
4, z=-7..-2,FillColor=[1,1,0,1]):
plot(quastrichp /*,Op, tp */, Scaling=Constrained):
```




```
quastrich
```


$$\left(6\cdot x^2 + 20\cdot\sqrt{3}\cdot x + 3\cdot y^2 + 4\cdot\sqrt{6}\cdot y + 9\cdot z^2 + 54\cdot\sqrt{2}\cdot z + 202\right)$$

Die Arbeitsweise ist dieselbe wie bei der Herstellung der Scheitelform einer Parabel.
Hier durch Hinsehen:

```
xterm:=hold(6*(x+5/3*sqrt(3))^2);expand(xterm);
yterm:=hold(3*(y+2/3*sqrt(6))^2);expand(yterm);
zterm:=hold(9*(z+3*sqrt(2))^2);expand(zterm);
```

$$6\cdot\left(x+\frac{5\cdot\sqrt{3}}{3}\right)^2$$

$$6\cdot x^2+20\cdot\sqrt{3}\cdot x+50$$

$$3\cdot\left(y+\frac{2\cdot\sqrt{6}}{3}\right)^2$$

$$3\cdot y^2+4\cdot\sqrt{6}\cdot y+8$$

$$9\cdot(z+3\cdot\sqrt{2})^2$$

$$9\cdot z^2+54\cdot\sqrt{2}\cdot z+162$$

Also

```
quastrichK:=xterm+yterm+zterm-18   //-220+202
```

$$9\cdot(z+3\cdot\sqrt{2})^2+6\cdot\left(x+\frac{5\cdot\sqrt{3}}{3}\right)^2+3\cdot\left(y+\frac{2\cdot\sqrt{6}}{3}\right)^2-18$$

```
quastrich-expand(quastrichK)
```

$$(\,0\,)$$

hier muss 0 herauskommen

Letzter Teil der Hauptachsentransformation ist die Translation t

```
t:=matrix([5/3*sqrt(3), 2/3*sqrt(6),3*sqrt(2)]);
tt:=linalg::transpose(t):
```

$$\begin{pmatrix} \frac{5\cdot\sqrt{3}}{3} \\ \frac{2\cdot\sqrt{6}}{3} \\ 3\cdot\sqrt{2} \end{pmatrix}$$

$\vec{p}\,''=\vec{p}\,'+\vec{t}$ also $\vec{p}\,'=\vec{p}\,''-\vec{t}$ Das ergibt:

```
quaH:=Simplify(expand((pt-tt)*Pt*A*P*(p-t)+at*P*(p-t)))+
202;
quadrikH:=ew1*x^2+ew2*y^2+ew3*z^2-18
```

$$\left(\,6\cdot x^2+3\cdot y^2+9\cdot z^2-18\,\right)$$

$$6\cdot x^2+3\cdot y^2+9\cdot z^2-18$$

$$6 \cdot x^2 + 3 \cdot y^2 + 9 \cdot z^2 - 18$$

Angabe der Gleichung in der üblichen Form:

```
hold(x^2/3+y^2/6+z^2/2=1)
```

$$\frac{x^2}{3} + \frac{y^2}{6} + \frac{z^2}{2} = 1$$

```
quadrikHp:=plot::Implicit3d(quadrikH,x=-2..2,y=-3..3,
z=-2..2,
Scaling=Constrained):
plot(%)
```



Bestimmung des ursprünglichen Mittelpunktes:

$$\vec{m}'' = \begin{pmatrix} 0 \\ 0 \end{pmatrix}, \quad \vec{m}' = -\vec{t}, \quad \vec{m} = P(-\vec{t})$$

nun in 3D

```
m:=Simplify(P*(-t))
```

$$\begin{pmatrix} 4 \\ -2 \\ -3 \end{pmatrix}$$

Dieses ist der alte Mittelpunkt.

Bestimmung des Urbildes des rechten Hauptscheitels:

$$\vec{r}'' = \begin{pmatrix} r \\ 0 \end{pmatrix}, \vec{r}' = \begin{pmatrix} r \\ 0 \end{pmatrix} - \vec{t}, \vec{r} = P\begin{pmatrix} r \\ 0 \end{pmatrix} + \vec{m}, \ r = r \cdot \overrightarrow{ev_1} + \vec{m}$$

Mit ev1 rechts ist hier der normierte 1. Eigenvektor gemeint. Alles jetzt in 3D


```
r:=sqrt(3): // Halbachse in (späterer) x-Richtung
rur:=r*ev1n+m;

rurstrich:=Pt*rur;
```

```
rurstrich:=Pt*rur;
```

$$\begin{pmatrix} 3 \\ -3 \\ -2 \end{pmatrix}$$

$$\begin{pmatrix} -\frac{2\cdot\sqrt{3}}{3} \\ -\frac{2\cdot\sqrt{2}\cdot\sqrt{3}}{3} \\ -3\cdot\sqrt{2} \end{pmatrix}$$

```
mp:=plot::Point3d(m,PointSize=2, PointColor=[0,1,0]):
Op:=plot::Point3d([0,0,0], PointSize=2, PointColor=[1,0,0]):
rurp:=plot::Point3d(rur, PointSize=2, PointColor=[0,1,1]):
rursp:=plot::Point3d(rurstrich, PointSize=2, PointColor=[1,0.5,0]):
rp:=plot::Point3d([r,0,0],PointSize=2, PointColor=[0,0,1]):
ev1urp:=plot::Arrow3d(m,m+3*ev1):
ev2urp:=plot::Arrow3d(m,m+3*ev2):
ev3urp:=plot::Arrow3d(m,m+3*ev3):
tp:=plot::Arrow3d(-t,[0,0,0],LineColor=[1,0,0]):
plot(quadrikp,mp,Op,rurp,tp,
      ev1urp,ev2urp,ev3urp,PointSize=2,Scaling=Constrained);
```



Test ob es eine Drehung allein oder gefolgt von Spiegelung ist. det=1 reine Drehung

```
linalg::det(Pt)
```

$$1$$

```
plot(quadrikp,quastrichp,
quadrikHp,mp,Op,tp,rp,rurp,rursp,

ev1urp,ev2urp,ev3urp,Axes=Origin,Scaling=Constrained)
```




##

Weitere Untersuchungen,

die Abbildung wird von Pt geleistet

```
Pt
```

$$\begin{pmatrix} -\frac{\sqrt{3}}{3} & -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \\ \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{3} \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \end{pmatrix}$$

Suche nach der Drehachse, sie muss die Eigenrichtung der Drehmatrix sein.

```
evliP:=linalg::eigenvectors(Pt)
```

$$\left[\left[1,\ 1,\ \left[\begin{pmatrix} -\frac{\sqrt{2}\cdot\sqrt{3}-\sqrt{3}}{2\cdot\sqrt{3}+3} \\ \frac{\sqrt{2}\cdot\sqrt{3}+3\cdot\sqrt{2}+\sqrt{3}}{2\cdot\sqrt{3}+3} \\ 1 \end{pmatrix}\right]\right],\ \left[\frac{\sqrt{2}\cdot\sqrt{3}}{12}-\frac{\sqrt{3}}{6}-\frac{\sqrt{\frac{2\cdot\sqrt{3}}{3}-\frac{\sqrt{2}}{3}-\frac{\sqrt{2}\cdot\sqrt{3}}{3}-\ldots}}{2}\ldots\right.\right.$$

```
float(evliP)
```

$$\left[\left[1.0,\ 1.0,\ \left[\begin{pmatrix} -0.1109881895 \\ 1.303225373 \\ 1.0 \end{pmatrix}\right]\right],\ \left[-0.5845509894-0.811356975\cdot i,\ 1.0,\ \left[\begin{pmatrix} 0.064\ldots \\ -0.761\ldots \end{pmatrix}\right.\right.\right.$$

Wie erwartet ist nur ein Eigenwert reell und zwar ist er 1.
Drehachsenrichtung:

```
dreh:=evliP[1][3][1]
```

```
dreh:=evliP[1][3][1]
```

$$\begin{pmatrix} -\frac{\sqrt{2}\cdot\sqrt{3}-\sqrt{3}}{2\cdot\sqrt{3}+3} \\ \frac{\sqrt{2}\cdot\sqrt{3}+3\cdot\sqrt{2}+\sqrt{3}}{2\cdot\sqrt{3}+3} \\ 1 \end{pmatrix}$$

Die Idee, dass die Normale auf der Ebene durch 0, M und M' die Drehachse sei, ist falsch:

```
mtt:=linalg::crossProduct(-t,m); float(mtt)
```

$$\begin{pmatrix} 2\cdot\sqrt{6}-\sqrt{2}\cdot 6 \\ -\sqrt{2}\cdot 12-\sqrt{3}\cdot 5 \\ \frac{10\cdot\sqrt{3}}{3}+\frac{8\cdot\sqrt{6}}{3} \end{pmatrix}$$

$$\begin{pmatrix} -3.586301889 \\ -25.63081679 \\ 12.30547534 \end{pmatrix}$$

Obiger Vektor stimmt nicht mit den Normalenvektor überein.

```
drehp:=plot::Arrow3d(5*dreh, LineColor=[1,0,1])
```

$$\text{plot::Arrow3d}\left([0,\ 0,\ 0],\ \left[-\frac{5\cdot(\sqrt{2}\cdot\sqrt{3}-\sqrt{3})}{2\cdot\sqrt{3}+3},\ \frac{5\cdot(\sqrt{2}\cdot\sqrt{3}+3\cdot\sqrt{2}+\sqrt{3})}{2\cdot\sqrt{3}+3},\ 5\right]\right)$$

Einzeichnen der Drehachse:

```
plot(quadrikp,quastrichp,
quadrikHp,mp,Op,tp,rp,rurp,rursp,

ev1urp,ev2urp,ev3urp,drehp,Axes=Origin,Scaling=Constrain
ed)
```



Die besondere Herausfordung, diese Drehung animiert zu gestalten, habe ich noch nicht umgesetzt.

nicht umgesetzt.
Zunächst muss ich herausbekommen, in welchen Ebenen sich die Punkte jeweils bewegen.
Hessesche Normalenform:

```
float(linalg::scalarProduct( (p-m),dreh)=0);
float(linalg::scalarProduct( (p+t),dreh)=0)
```

$$1.303225373 \cdot y - x \cdot 0.1109881895 + 1.0 \cdot z + 6.050403504 = 0.0$$

$$1.303225373 \cdot y - x \cdot 0.1109881895 + 1.0 \cdot z + 6.050403504 = 0.0$$

Hier sieht man, dass tatsächlich m und mstrich=-t in dieser Ebene liegen.
Wo durchstößt die Drehachse diese Ebene?
Dieser Durchstoßpunkt muss das k-fache des Drehachsenvektors sein.)

```
1.303225373^2*k - 0.1109881895^2*k + 1.0^2*k +
6.050403504 = 0.0;
solve(%)
```

$$2.686077995 \cdot k + 6.050403504 = 0.0$$

$$\{[k = -2.252504773]\}$$

```
delete k:
```

Der Durchstoßpunkt ist also

```
k:= -2.252504773: dm:=float(k*dreh)
```

$$\begin{pmatrix} 0.2500014267 \\ -2.935521373 \\ -2.252504773 \end{pmatrix}$$

Bestimmung des Drehwinkels

```
m-dm, -t-dm, linalg::scalarProduct(m-dm,-t-dm)
```

$$\begin{pmatrix} 3.749998573 \\ 0.9355213726 \\ -0.747495227 \end{pmatrix}, \begin{pmatrix} -\frac{\sqrt{3} \cdot 5}{3} - 0.2500014267 \\ 2.935521373 - \frac{\sqrt{6} \cdot 2}{3} \\ 2.252504773 - \sqrt{2} \cdot 3 \end{pmatrix}, -9.056658346$$

```
cosi:=linalg::scalarProduct(m-dm,-t-dm)/
sqrt(linalg::scalarProduct(m-dm,m-dm)*
linalg::scalarProduct(-t-dm,-t-dm))
```

$$-0.5844348207$$

```
float(arccos(cosi))
```

$$2.194979696$$

##
Das mache ich jetzt zur Probe auch für den rechten Hauptscheitel

```
rurs:=rurstrich:
```

```
rurs:=rurstrich:
float(linalg::scalarProduct( (p-rur),dreh)=0);
float(linalg::scalarProduct( (p-rurs),dreh)=0)
```

$$1.303225373 \cdot y - x \cdot 0.1109881895 + 1.0 \cdot z + 6.242640687 = 0.0$$

$$1.303225373 \cdot y - x \cdot 0.1109881895 + 1.0 \cdot z + 6.242640687 = 0.0$$

Hier sieht man, dass tatsächlich rur und rurstrich=rurs in derselben Ebene liegen.
Wo durchstößt die Drehachse diese Ebene?
Dieser Durchstoßpunkt muss das k-fache des Drehachsenvektors sein.)

```
delete kk:
1.303225373^2*kk - 0.1109881895^2*kk + 1.0^2*kk
+6.242640687 = 0.0;
solve(%)
```

$$2.686077995 \cdot kk + 6.242640687 = 0.0$$

$$\{[kk = -2.324072756]\}$$

Der Durchstoßpunkt ist also

```
kk:= -2.324072756: rm:=float(kk*dreh)
```

$$\begin{pmatrix} 0.2579446275 \\ -3.028790584 \\ -2.324072756 \end{pmatrix}$$

```
rur-rm, rurs-rm, linalg::scalarProduct(rur-rm,rurs-rm)
```

$$\begin{pmatrix} 2.742055372 \\ 0.02879058395 \\ 0.324072756 \end{pmatrix}, \begin{pmatrix} -\frac{\sqrt{3} \cdot 2}{3} - 0.2579446275 \\ 3.028790584 - \frac{\sqrt{2} \cdot \sqrt{3} \cdot 2}{3} \\ 2.324072756 - \sqrt{2} \cdot 3 \end{pmatrix}, -4.455121041$$

```
cosi:=linalg::scalarProduct(rur-rm,rurs-rm)/
       sqrt(linalg::scalarProduct(rur-rm,rur-rm)*
       linalg::scalarProduct(rurs-rm,rurs-rm))
```

$$-0.5842996474$$

```
wradm:=float(arccos(cosi));
wgrdm:=float(wradm/PI*180);
```

$$2.194813122$$

$$125.7535287$$

Das sind dieselben Werte wie für m und mstrich

Das sind dieselben Werte wie für m und mstrich

```
wradm;
wgrdm;
```

2.194813122

125.7535287

Damit handelt es sich also um eine Drehung um 125,75.. Grad.